
US Training Days

Our US training days will take place on the 13" and 14™ of November at the Doubletree
Hotel, Washington DC, prior to the Society for Neuroscience annual meeting. The training
days are suitable for both existing and prospective users of Spike2 and Signal and
include introductions to the software along with sessions on script writing, stimulus 5
generation and on-line control. Our popular workshop thread, where you can discuss
your questions and requirements with a CED engineer or programmer in a small group,

will also be available.

If you or any of your colleagues are interested in attending, please follow the links to view =
the training day program and complete the registration form.

Meetings and events

Society for Neuroscience
Washington Convention Center
Washington, DC, USA
November 15" — 19" 2008

British Pharmacological Society Winter Meeting
Hilton Brighton Metropole

Brighton, UK

December 16" — 18™ 2008

Australian Neuroscience Society
National Convention Centre
Canberra, Australia

January 27" — 30" 2009

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED downloads page, or by
clicking on the links in the table below. Demonstration versions of the latest software are
also available.

Spike2 downloads Signal downloads
Spike2 version 6.10 Signal version 4.02
Spike2 version 5.20 Signal version 3.10
Spike2 demo Signal demo

Q. How do | go about running multiple synchronised 1401’'s on a single PC with
Spike2?
A. Multiple synchronised 1401 interfaces can be run on a single PC to record many

channels at high sampling rates. Each 1401 is controlled by a separate running


http://www.ced.co.uk/nw9y.htm
http://www.ced.co.uk/tdapply.htm
http://www.sfn.org/am2008/
http://www.bps.ac.uk/site/cms/contentCategoryView.asp?category=258#WIN08
http://www.sallyjayconferences.com.au/ans2009/index.html
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigdemu.shtml

copy of Spike2 and the data is saved to a separate data file per copy of Spike2.
You should set the sampled data to save to different locations in the Automation
tab of each sampling configuration. Files recorded with separate 1401's can be
combined off-line (up to a maximum single file size of 2GB) using the
FileMerge.s2s script. This is included in the Scripts folder, which can be found
in the Spike2 directory on your PC.

To open multiple copies of Spike2 and assign them to individual 1401’'s the
command line of Spike2 needs to be edited, which can be achieved easily via the
Spike2 desktop shortcut icon. Whenever you double-click the Spike2 shortcut, it
automatically checks to see if there is already a running copy open and, if so, the
new one quits. You can remove this check for a running copy by adding the
character /M to the end of the command line for the Spike2 program. The
command line can be accessed by right-clicking on the Spike2 shortcut icon from
the desktop and selecting the shortcut tab in the dialog.

You can also specify which connected 1401 the new copy of Spike2 should
control. This is done by adding /Un to the end of the command line, where n is a
number between 1 and 8. To make sure that the new Spike2 uses the lowest
numbered available 1401 the default /UO is used. Each time Spike2 is run, a new
copy is opened and assigned to the next available connected 1401.

Signal

Q.

It would be very useful when running LTP experiments to be able to specify
different sweep lengths for responses sampled under different stimulus
conditions. Is this possible?

Multiple states can be defined with variable sweep points when using Outputs
frame and Fixed interval sampling mode. The option for variable sweep points
can be enabled in the General tab of the Signal sampling configuration once the
Multiple frame states checkbox is selected. The number of sampled points to use
per state can then be set in the ADC points field of the Pulses configuration
dialog.

When recording many channels with high sample rates you can set-up Spike2 to record multiple data files in sequence from the Automation tab of the sampling configuration

Scripts: Spike2 @L

Q. | have be ing Spike2 and a Micro1401 to play short stimulus waveforms and
record responses, but in our next series of experiments | need to play out a long
(10-20 mins) continuous stimulus waveform. As the Micro1401 has the option of
1 or 2MB internal memory, is there any way to cut these long waveforms into
sections and then cycle through the consecutive sections (loading each section
as it's needed) to give the impression of continuous output?

A. The attached script OutputLongWave.s2s can be used to output long
waveforms by loading sections of the source waveform into 1401 memory in-turn.
The script generates a toolbar from where the user can select the Spike2 file to
use for the source waveform. The script then uses the current sampling
configuration to load in the sections of waveform, which is output from DAC 0 of
the 1401.

Scripts: Signal J@
Q. I would li display a data file recorded in Signal as a continuous waveform for
publication, rather than a series of frames. Is this possible?

A. The attached script, MergeToSingleFrame.sgs will produce a new data file of
up to 2GB with all of the frames from the source file merged into a single frame of
continuous data.

Did you know...?

When you apply a digital filter to create a new channel in Spike2, the channel comment
holds the source channel number and a description of the filter.

Recent questions

Q. | have used the graphical editor in Spike2 to setup a sinewave output, but | would
like this to be triggered in response to an event input during a recording. How can
| do this?

A. The output sequencer has a number of commands that can be used to control

how the sequence proceeds and how it reacts to incoming data or external
triggers. The bottom row of icons in the graphical editor ‘pulse palette’ include
wait and branch commands that can be dragged onto the control track in the
main window. The wait command can be used to generate random wait times or
monitor channel data until a certain condition is met. From the drop down list of
the wait icon, select Next event and then set the Channel field to the event
channel being used to record the trigger pulse.

Back to top

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

You can give each state a label, which will then be displayed on the corresponding button on the state toolbar


'$OutputLongWave|Script to load long waveforms into memory in sections and replace

'with new during recording to give continuous output



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 25/10/08



Var work%[50000];        'Array to hold waveform sections 

var data%;                 'Data file containing waveform to output

var chan%:=1;              'Channel to output

Var length;                 'Length of time for each data section

Var Start:=0;               'Start time

Var EndT;                   'End time

Var go%;                    'Flag to start...

Var Done%;                  '...and stop output

var nFile%;                        'Sampled data file

var pStart,pEnd,lEnd;           'Times of currenlty playing and loaded wavefom sections



DoToolbar();   'Try it out

Halt;



Func DoToolbar()    'Set your own name...

ToolbarClear();     'Remove any old buttons

ToolbarSet(0, "", Idle%);   'Idle routine

ToolbarSet(1, "&Quit");    'This button quits the script

ToolbarSet(2, "New File", File%);   'Link to File select

ToolbarSet(3, "Start", Start%);     'Start sampling and output

ToolbarEnable(3,0);     'Disable Start button 

return Toolbar("Script to output long waveforms", 1023);

end;



Func Idle%()   'Idle routine for output on-line

var key$;

Var num%, Err%;

If data%>0 then

Key$:=PlayWaveStatus$();

If Key$="B" and Done%=0 then

   Done%:=1;							'Know the first time to second areas

	Seconds(0);

Endif



If Key$="B" and done%=1 then 'Check if we've done this already

    View(data%);

        pStart := Start-length;     'Get start and end times

        pEnd := Start;

        lEnd := EndT;

	ArrConst(work%[], 0);       'Initialise work array

   Num%:=ChanData(Chan%,Work%[],Start,Endt);    'get next section of waveform

   if num%>0 then

	   Err%:=PlayWaveCopy("A",Work%[],0);   'Add it to the play wave area

	   Printlog("\nUpdated A %d %d at %g",num%,Err%, seconds());    'Print current section details

            PrintLog("\nCurrent %g %g %S",Start,EndT,Key$);

      Start:=EndT;

            EndT:=Start+length;

                PrintLog("\nNew= %g %g %S",Start,EndT,Key$);    

                Cursor(1,pStart);

    Cursor(2,pEnd);

    Cursor(3,lEnd);

      done%:=2;

	else								' No more data, cancel the link

	   PlayWaveLink$(Key$, Chr$(0));        'Link to next playwave section

   Endif

endif

If Key$="A" and Done%=2 then    'Repeat as above for next section

        View(data%);

                pStart := Start-length;

        pEnd := Start;

        lEnd := EndT;

	ArrConst(work%[], 0);

   Num%:=ChanData(Chan%,Work%[],Start,Endt);

   if num%>0 then

	   Err%:=PlayWaveCopy("B",Work%[],0);

		Printlog("\nUpdated B %d %d at %g",num%,Err%,seconds());

		PrintLog("\nCurrent%g %g %S",Start,EndT,Key$);

      Start:=EndT;

      EndT:=Start+length;

            PrintLog("\nNew=%g %g %S",Start,EndT,Key$);

                            Cursor(1,pStart);

    Cursor(2,pEnd);

    Cursor(3,lEnd);

      done%:=1;

	else								' No more data, cancel the link

	   PlayWaveLink$(Key$, Chr$(0));        'Loop back to start

   Endif

Endif

Endif



return 1;  'This leaves toolbar active

end;







Func File%()   'Open a file for output

var Err$;

Start:=0;

EndT:=0;

Data%:=FileOpen("",0,1,"Data file to play");        'Prompt user to select a file

if data% > 0 then

    Window(0,0,100,50);     'Display it at the top

    CursorSet(3);               'Set cursors to mark areas

    CursorLabel(4,1,"Playing");

    CursorLabel(4,2,"Loaded");

    CursorLabel(4,3,"");

    length:=50000/(1/BinSize(chan%));   'Get length of time of individual waveform sections

    EndT:=Start+length;

    pStart := Start;

    pEnd := EndT;

    PlayWaveAdd("A","First",0,1/BinSize(chan%),50000);      'Add areas of correct length to playwave sections

    PlayWaveAdd("B","Second",0,1/BinSize(chan%),50000);

    View(data%).PlayWaveAdd("A","First",0,Start,EndT,chan%);       'Add first section



    Start:=EndT;

    EndT:=Start+length;

    View(data%).PlayWaveAdd("B","Second",0,Start,EndT,chan%);   'Add second section

    lEnd := EndT;

    Err$ := PlayWaveLink$("A", "B");    'Link play wave areas so that they loop

    Err$ := PlayWaveLink$("B", "A");

    Start := EndT;

    EndT := Start+length;

    Cursor(1,pStart);       'Position cursors in original file to show current and next position of output

    Cursor(2,pEnd);

    Cursor(3,lEnd);

    done% := 0;

    nFile% := FileNew(0,2);     'Open new data file

    if nFile% > 0 then

        Window(0,50,100,100);   'Display at bottom of screen

        FrontView(nFile%);

        ToolbarEnable(3,1);     'Enable Start button on toolbar

    endif;

endif;

return 1;  'This leaves toolbar active

end;







Func Start%()

View(nFile%);

SampleStart();      'Start sampling

Yield();                'Wait one cyle time

SampleKey("A");     'Start waveform output

ToolbarSet(3, "Stop",Stop%);      'Set Stop button

return 1;

end;





Func Stop%()

View(nFile%);

SampleStop();       'Stop sampling

ToolbarSet(3, "Start", Start%);     'Set Start button

ToolbarEnable(3,0);     'Disable Start button

return 1;

end;







CED
OutputLongWave.s2s


'$MergeToSingleFrame|Script to produce a new file with all frames merged



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 25/10/08



'---------Declare global variables-----------

var oldFile%;

var newFile%;									'Handle for downsampled data file

var oldSize;									'Bin size in source file

var newSize;									'Bin size in new file

var points%;									'Number of data points in source file frames

var newPoints%;									'Number of points in downsampled file

var factor%;									'Down samping factor	

var numChans%;									'Number of channels

var chan%,frame%;								'Counters for loops

var numFrames%;									'Number of frames



View(App(3)).WindowVisible(0);

'---------Set toolbar buttons--------------



ToolbarSet(1,"Quit");

ToolbarSet(3,"Load source file",Load%);

ToolbarSet(5,"Merge",DoMerge%);

ToolbarSet(7,"Save",DoSave%);

ToolbarEnable(5,0);

ToolbarEnable(7,0);

Toolbar("Merge to single frame script",1023);





Func Load%()

if oldFile% > 0 then 							'If existing file opened

	View(oldFile%);

	FileClose(-1);								'Close existing file	

endif;

oldFile%:=FileOpen("",0,0);						'Open new source file

if oldFile% < 0 then							'Check file opened

	Message ("Failed to open file");

endif;

Window(0,0,100,100);							'Position window

FrontView(oldFile%);							'Bring to front

if oldFile% >0 then

	ToolbarEnable(5,1);

endif;

return 1;

end;





Func DoMerge%()									

View(oldFile%);

points%:=ChanPoints(1);				'Get number of data points in source file

numChans%:=ChanCount(-4);						'Count waveform channels

numFrames%:=FrameCount();							

newPoints%:=points% * numFrames%;					'Get number of data points for new file



newFile%:=SetMemory(numChans%,newPoints%,View(oldFile%).BinSize(1),View(oldFile%).MinTime(),0,0,0,"Merged","s");	

												'Create empty data file with frame length set to total length of data in original file 

Window(50,0,100,100);

DownSample();

return 1;

end;





Proc DownSample()								'Procedure to loop through frames and channels

var countF%;									'Counter to loop through each frame

var countC%;									'Counter to loop through each channel

for countF%:=1 to numFrames% do					'For each frame

	View(oldFile%).Frame(countF%);				'Ensure correct frame is viewed	

	for countC%:=1 to numChans% do				'For each channel	

		DSChan(countC%,countF%-1);				'Link to procedure for downsampling channel

	next;

next;

View(oldFile%);Window(0,0,50,100);				'Position window

FrontView(oldFile%);					

FrontView(newFile%);

ToolbarEnable(7,1);

end;





Proc DSChan(chan%,st%)

var count%;											'Counter to loop through data points

var num%:=0;



ArrConst(View(newFile%,chan%).[st%*points%:points%],View(oldFile%,chan%).[]);		'Copy array to new file

View(newFile%).Optimise(-1);						'Optimise data 

end;



Func DoSave%()

var goodSave%;

View(newFile%);

goodSave%:=FileSaveAs("",0,"Save new file");

if not goodSave% then ToolbarEnable(7,0)endif;

return 1;

end;

CED
MergeToSingleFrame.sgs

If you have any comments about the newsletter format and content, or wish to un- Fack o fop
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

